
3 Liste ı̂n ML

O listă e un s, ir ordonat, finit, de elemente de acelas, i tip.
În ML, putem scrie valori de tip listă ı̂ncadrate ı̂ntre paranteze drepte [ s, i ], cu elementele separate

prin punct-virgulă: [9; 8; 10; 9] , ["ana"; "are"; "mere"] .

3.1 Construirea s, i descompunerea listelor

O listă poate fi definită recursiv: e fie lista vidă, fie un element urmat de o listă. În limbajele de
programare, listele modelează deobicei această definit, ie: pentru a ajunge la un anumit element, lista
trebuie parcursă pornind de la ı̂nceput (spre deosebire de vectori, care oferă acces direct la un element
cu un anumit indice).

Pentru a lucra cu liste conform acestei definit, ii recursive, e nevoie să
• construim o listă vidă: [] (indiferent de tipul elementelor)
• construim o listă dintr-o listă existentă s, i un nou element (care devine capul listei, vechea listă

devenind coada celei noi). Aceasta se face cu constructorul :: cu sintaxa de operator binar infix:
cap :: coada . De exemplu, 1::[] e lista [1], iar 1::2::[3;4] e lista [1;2;3;4] .

Atent, ie! Operatorul :: produce o listă, nu e nevoie să folosim paranteze drepte ı̂n jurul rezultatului.
Astfel 1::[2;3] ı̂nseamnă [1;2;3] pe când [1::[2;3]] ı̂nseamnă [[1;2;3]], deci o listă cu un singur
element, care e la rândul lui o listă de trei ı̂ntregi. Deasemenea, :: nu poate lega două liste, ci doar
un element de o listă (cu elemente de acelas, i tip). [1;2]::[3] e incorect (nu compilează)!

Exercit, iul 1. Scriet, i o funct, ie care returnează lista cifrelor unui număr natural dat ca parametru.
Solut, ie. Privim numărul recursiv, ca un s, ir de cifre: fie o singură cifră, fie o cifră (ultima)

precedată de alt număr. Cum cifra pe care o extragem dintr-un număr e ultima, iar ı̂ntr-o listă,
elementul distinctiv e primul, cel mai us,or e să construim lista cifrelor ı̂n ordine inversă:
let rec revdiglst n = if n < 10 then [n] else (n mod 10) :: revdiglst (n/10)

În cazul de bază, se creează o listă cu o cifră (chiar n). În cazul recursiv, revdiglst (n/10) va
produce lista cifrelor primei părt, i ı̂n ordine inversă, la care se adaugă ultima cifră, acum ı̂n capul listei.
Cum apelul de funct, ie e mai prioritar decât operatorul :: nu e nevoie de paranteze ı̂n jurul cozii listei.

Putem obt, ine cifrele ı̂n ordinea init, ială folosind un rezultat part, ial. Redefinim exprimarea recursivă
a problemei ca funct, ie cu doi parametri: creează o listă cu cifrele unui număr, având lista res a ultimelor
cifre deja parcurse, s, i numărul n rămas ı̂n fat,a lor. Pentru numărul init, ial 1457, la un moment dat
avem lista res=[5;7] s, i a mai e de prelucrat n=14. Atunci, ultima cifră a lui n trebuie pusă ı̂n capul
listei res, continuând recursiv până la epuizarea cifrelor din n, când res e chiar rezultatul dorit.

Cum numărul are măcar o cifră, putem porni apelul recursiv cu rezultatul part, ial [n mod 10],
numărul rămas fiind n / 10. Deci vom avea let diglst n = diglst2 [n mod 10] (n / 10) s, i a
rămas să scriem (̂ınainte de aceasta) funct, ia recursivă diglst2. Dacă numărul transmis e zero, cifrele
s-au terminat (cazul de bază) s, i returnăm lista acumulată, altfel ultima cifră trebuie pusă ı̂n fat,a listei
existente, continuând cu numărul rămas ı̂n fat, ă (n / 10). Scriem deci ı̂n fat,a definit, iei lui diglst:
let rec diglst2 res n = if n = 0 then res else diglst2 ((n mod 10)::res) (n/10)

Această funct, ie fiind doar ajutătoare, o putem scrie ca declarat, ie locală ı̂n cadrul funct, iei dorite:
let diglst n =

let rec diglst2 res n = if n = 0 then res else diglst2 ((n mod 10)::res) (n/10)
in diglst2 [n mod 10] (n / 10)

3.2 Potrivirea de tipare

După definit, ia recursivă, o listă e fie lista vidă, fie un element (capul) urmat de o altă listă (coada).
Pentru a putea scrie funct, ii care lucrează cu liste, avem nevoie să distingem ı̂ntre aceste două cazuri,
iar ı̂n al doilea, să descompunem o listă ı̂n elementele componente (cap s, i coadă).

În limbajele precum cele funct, ionale, aceasta se face printr-un mecanism numit potrivire de tipare
(pattern matching), prin care se determină dacă structura unei valori (obiect) corespunde unui tipar dat,
permit, ând ı̂n acest caz identificarea s, i numirea părt, ilor componente, s, i apoi folosirea lor ı̂n prelucrări.

Sintaxa potrivirii de tipare e

match expr0 with
| tipar1 -> expr1
| tipar2 -> expr2
...

Întreaga construct, ie e o expresie. Dacă valoarea obt, inută prin evaluarea lui expr0 se potrives,te cu
structura indicată de tipar1, valoarea ı̂ntregii expresii e expr1 ; altfel, dacă se potrives,te cu structura
lui tipar2, valoarea rezultantă e dată de expr2, etc. Bara verticală | pentru prima variantă de tipar e
opt, ională, dar deobicei se scrie, pentru o formatare consistentă a codului.

Cel mai simplu tipar e o valoare constantă: putem astfel scrie pentru negat, ia booleană:
let neg b = match b with

| false -> true
| true -> false

Putem identifica orice alte valori structurate: perechi, n-tuple, liste, tipuri definite de utilizator.
Pot fi folosite două sau mai multe tipare cu rezultat comun, separându-le tot cu bara |

let rec cmmdc a b = match (a, b) with
| (a, 0) | (0, a) -> a (* tipare cu o componenta constanta zero *)
| (a, b) -> cmmdc b (a mod b) (* orice alt caz *)

În scrierea tiparelor sunt valabile următoarele reguli:
• Pe fiecare variantă (ramură), toate numele folosite ı̂n stânga lui -> sunt identificatori nou definit,i

pentru a desemna părt, ile componente ale tiparului, s, i domeniul lor de vizibilitate e partea dreaptă
(valoarea rezultantă). Chiar folosind nume existente, ele capătă ı̂n cadrul tiparului un alt ı̂nt,eles.
(Mai sus, identificatorul a din tipar e legat de o componentă a perechii, nu e parametrul funct, iei.)

• Un identificator poate apare o singură dată ı̂ntr-un tipar, s, i se potrives,te cu orice valoare (cu
structură oricât de simplă/complexă).

• Identificatorul special _ (linia de subliniere) e singurul care poate apare de mai multe ori ı̂ntr-un
tipar (fără a implica vreo egalitate ı̂ntre fragmentele structurale pentru care e folosit).

• Potrivirea de tipare se testează pe rând. Deci, pentru tipare cu structură comună, ordinea scrierii
contează: o variantă de tipar e aplicabilă doar dacă nu s-au potrivit nicicare din tiparele anterioare.

• Compilatorul avertizează dacă există variante nefolosite sau lipsă (tipare netratate), aceste verificări
fiind un mare avantaj al mecanismului de potrivire de tipare (eliminând erorile la rulare).
O scriere simplificată se poate folosi pentru potrivirea de tipare ı̂n argumentul unei funct, ii:
let numefct alte arg = function

| tipar1 -> expr1
| tipar2 -> expr2
...

În acest caz, nu se mai dă nume pentru ultimul (posibil unicul) argument al funct, iei, deoarece ı̂n
fiecare variantă, el va fi identificat prin numele date ı̂n acele tipare. De exemplu, funct, ia semn:
let sgn = function

| 0 -> 0
| x -> if x > 0 then 1 else -1

Ordinea variantelor contează aici, pentru că identificatorul x se potrives,te cu orice tipar; deci,
scriind ı̂n ordine inversă, s, i pentru valoarea 0 s-ar returna -1 (dar compilatorul avertizează ca tiparul
0 nu se va activa vreodată).

3.2.1 Tipare pentru liste

În tiparele pentru liste, putem folosi lista vidă s, i constructorul :: pentru liste. Cel mai frecvent
distingem cele două cazuri: lista vidă sau nu. De exemplu, suma elementelor unei liste de ı̂ntregi:
let rec sumlist = function

| [] -> 0
| h :: t -> h + sumlist t

Numele h s, i t sunt alese de utilizator, sugerând capul listei (head) s, i respectiv coada (tail).
Putem folosi s, i tipare mai complicate, de exemplu pentru a afla dacă primele două elemente sunt egale:

let equal12 = function
| e1 :: e2 :: _ when e1 = e2 -> true
| _ -> false

Potrivirea de tipare poate distinge doar structura unei valori (sau o constantă), s, i nu egalitate,
deci nu se putea scrie e :: e :: _ -> true (compilatorul ar fi semnalat eroare pentru folosirea unui
identificator de două ori ı̂ntr-un tipar). Limbajul permite ı̂nsă adăugarea unei clauze when condit,ie,
pentru a impune condit, ii elementelor unui tipar. Linia de subliniere e folosită pentru (sub)tiparele
irelevante (coada listei, respectiv al doilea tipar, care acoperă restul cazurilor: listele cu ≤ 1 element,
s, i cele cu primele două elemente distincte. Echivalent se putea scrie mai scurt:
let equal12 = function
| e1 :: e2 :: _ -> e1 = e2
| _ -> false

În acest caz, primul tipar identifică listele cu ≥ 2 elemente (cu rezultatul boolean dat de comparat, ie),
iar al doilea tipar, restul (listele cu ≤ 1 element).

Pentru a determina dacă o valoare apare ı̂ntr-o listă, putem scrie că un element e ı̂n listă dacă e fie
capul listei, fie apare ı̂n coada listei (operatorul || ı̂nseamnă SAU ):
let rec mem x = function

| [] -> false
| h :: t -> h = x || mem x t

Funct, ia are doi parametri, valoarea căutată (x), s, i lista (pe care folosim potrivirea de tipare cu
function). Nu am fi putut scrie un tipar x :: t -> true, pentru că x din tipar ar fi un identificator
nou reprezentând capul listei, fără legătură cu parametrul x (la care atunci nu ne-am mai putea referi).
Funct, ia mem există s, i ca funct, ie definită ı̂n modulul standard List pentru lucru cu liste.

3.3 Funct, ii predefinite pentru liste

Funct, iile predefinite pentru liste sunt grupate ı̂n modulul List. Le folosim deci cu numele complet
List.numefunct,ie. (Dacă am deschide modulul cu directiva open List s-ar putea folosi s, i numele
simple, dar se preferă numele complet, pentru a fi mai clar ı̂n program unde se lucrează cu liste.)

Două funct, ii de bază, List.hd s, i List.tl permit obt, inerea capului s, i cozii unei liste. Dacă argu-
mentul e lista vidă, ele es,uează cu o except, ie, după cum putem verifica ı̂n interpretorul OCaml:
# List.hd [];;
Exception: Failure "hd".

Aceasta ı̂nseamnă că ı̂nainte de folosirea lor ar trebui să verificăm s, i să tratăm cazul listei vide.
Preferăm lucrul prin potrivire de tipare, deoarece compilatorul verifică automat dacă am tratat toate
cazurile; el nu poate detecta ı̂nsă că nu am comparat o listă cu [] ı̂nainte de a-i folosi elementele.

Putem implementa noi ı̂ns, ine orice funct, ie standard din modulul List, pentru a ı̂nt,elege cum
funct, ionează. De exemplu, funct, ia length pentru lungimea unei liste se poate scrie recursiv:
let rec length = function

| [] -> 0
| _ :: t -> 1 + length t

Deci, List.length nu e o operat, ie atomică, ci necesită parcurgerea ı̂ntregii liste (ignorând ı̂nsă
valoarea elementelor, după cum arată folosirea lui _ ı̂n tipar). Va fi deci costisitoare pentru liste lungi.

3.4 Recursivitatea finală (tail recursion)

Am folosit recursivitatea pentru a scrie solut, ii cât mai simple. Trebuie să ne ı̂ntrebăm s, i cât de eficiente
s, i utilizabile sunt funct, iile scrise. Să scriem progresia aritmetică cu baza 1 s, i rat, ia 2 ca s, ir recurent:
let rec arit n = if n = 0 then 1 else 2 + arit (n-1)

Apelând ı̂n interpretorul OCaml pe un calculator tipic arit 1000000 (un milion), obt, inem:
Stack overflow during evaluation (looping recursion?).

Observăm că valoarea funct, iei se poate calcula doar după revenirea din apelul recursiv arit (n-1),
abia atunci se face suma cu 2. La fiecare apel recursiv, programul trebuie să salveze ı̂n memorie locul

unde va relua calculul după revenirea din apel, s, i orice valori (parametri, definit, ii locale) necesare ı̂n
continuare. Valorile sunt salvate ı̂ntr-o zonă numită stivă, deoarece ele se vor folosi din nou ı̂n ordinea
inversă ı̂n care au fost plasate ı̂n memorie (primul apel recursiv e cel din care se revine ultimul). La
un număr prea mare de apeluri, memoria din stivă se epuizează, s, i programul es,uează.

Observăm că pe ramura recursivă, sigur vom aduna 2 la rezultatul apelului. Putem anticipa
adunarea dând funct, iei ı̂ncă un parametru r, ı̂n care acumulăm valoarea care trebuie adunată. Ajuns, i
la cazul de bază, adunăm la termenul de bază (1) valoarea acumulată (r) s, i obt, inem rezultatul final.
let rec arit2 r n = if n = 0 then 1 + r else arit2 (r + 2) (n - 1)

Init, ial, valoarea rezultatului acumulat e 0 (nu am adunat nimic). Putem scrie deci o funct, ie arit1
care apelează arit2 cu valoarea init, ială potrivită (deci arit1 va fi o funct, ie cu un parametru):
let arit1 =

let rec arit2 r n = if n = 0 then 1 + r else arit2 (r + 2) (n - 1)
in arit2 0

Pe ramura recursivă, rezultatul e returnat direct, nu mai sunt necesare alte calcule. Numim situat, ia
recursivitate finală (sau prin revenire; engl. tail recursion), pentru că apelul recursiv e ultima operat, ie
de efectuat pe ramura respectivă. Rezultatul e returnat neschimbat din fiecare apel până la cel init, ial,
sau (̂ın codul optimizat generat de compilator), chiar direct: nu mai sunt necesare informat, ii pe stivă
(adrese de revenire, parametri sau alte valori). Recursivitatea e convertită de compilator ı̂n iterat, ie.

Rescriem asemănător lungimea listei, numărând elementele ı̂ntr-un parametru acumulator:
let length lst =

let rec len2 r = function
| [] -> r
| _ :: t -> len2 (r+1) t

in len2 0 lst

Funct, ia nu mai e limitată de dimensiunea stivei s, i poate lucra cu argumente mari.

3.5 Parcurgerea cu funct, ii standard

De regulă, listele trebuie parcurse, cu diverse prelucrări pe elemente. Cum prelucrările sunt funct, ii, s, i
ı̂ntr-un limbaj funct, ional putem transmite funct, ii ca parametri, e natural să folosim funct, ii standard de
parcurgere, la care trebuie doar să specificăm prelucrarea dorită. Fără a mai trebui să scriem codul de
parcurgere, avem mai put, in cod, mai simplu s, i mai us,or de ı̂nt,eles, s, i cu mai put, ine riscuri de eroare.

Distingem trei mari categorii, după cum prelucrarea
• face ceva cu fiecare element, fără a returna o valoare (ex. tipăres,te). Folosim funct, ia List.iter .
• transformă fiecare element al listei, generând o nouă listă de aceeas, i lungime. Folosim List.map .
• combină valorile din listă (de la cap sau de la coadă). Folosim List.fold_left s, i List.fold_right.

3.5.1 List.iter

Tipul funct, iei List.iter e (’a -> unit) -> ’a list -> unit. Aici, ’a denotă o variabilă de tip,
care poate fi substituită cu orice tip. Tipul unit e un tip cu o singură valoare, notată () . E folosit
pentru funct, iile al căror scop nu e să producă o valoare utilă, ci un efect, cum ar fi de exemplu funct, iile
de tipărire. (̂In C, funct, iile scrise doar pentru efectul lor, fără a returna ceva, au tipul void). List.iter
ia o funct, ie cu domeniu de definit, ie arbitrar ’a s, i rezultat unit, s, i o listă cu acelas, i tip de elemente ’a
s, i aplică funct, ia fiecărui element al listei. Ea returnează unit (altfel spus, nu produce un rezultat).

Schematic:
[ e1 ; e2 ; ... en ]

f f ... f

List.iter Printf.printf "%d "[1;2;3] va tipări 1 2 3 cu un spat, iu (indicat de formatul "%d ")
după fiecare număr, inclusiv ultimul. List.iter print_int [1;2;3] va tipări (fără spat, ii) 123
(print_int e o funct, ie standard cu tipul int -> unit, care tipăres,te un ı̂ntreg).

3.5.2 List.map

Tipul funct, iei List.map e (’a -> ’b) -> ’a list -> ’b list. Deci, List.map ia ca parametru
o funct, ie arbitrară (cu parametru de tip ’a s, i rezultat de tip ’b), s, i o listă cu elemente de tip ’a,
aplicând funct, ia fiecărui element al listei, s, i returnând lista rezultatelor (o listă de aceeas, i lungime cu
cea init, ială, cu elemente de tip ’b. Schematic (notând cu f funct, ia aplicată):

[ e1 ; e2 ; ... en ]

f f ... f

[ r1 ; r2 ; ... rn ]

Exemple:
List.map (fun x -> x+1) [1;2;3] dă lista [2;3;4]
echivalent: List.map ((+) 1) [1;2;3]
List.map String.length ["ana";"are";"mere"] dă lista [3;3;4]
(funct, ia length din modulul String returnează lungimea unui s, ir)

3.5.3 List.fold_left s, i List.fold_right

Funct, iile List.fold_left s, i List.fold_right sunt cele mai versatile funct, ii pentru lucru cu liste. Ele
permit calculul unei valori de tip arbitrar din toate elementele listei, pornind de la o valoare init, ială s, i
aplicând la fiecare pas o funct, ie dată elementului curent s, i rezultatului obt, inut până ı̂n acel moment.
List.fold_left are tipul (’a -> ’b -> ’a) -> ’a -> ’b list -> ’a, deci are trei parametri: o
funct, ie cu doi parametri, de tip ’a s, i ’b s, i rezultat de tip ’a; o valoare init, ială de tip ’a, s, i o listă
de elemente de tip ’b; rezultatul e de acelas, i tip ’a ca s, i valoarea init, ială. Cu notat, iile din schema:

[ e1 ; e2 ; ... en ]

r0 f r1 f r2 ... rn−1 f rn

List.fold_left calculează succesiv valorile r1 = f(r0, e1), r2 = f(r1, e2), ..., rn = f(rn−1, en−1),
rezultatul final fiind rn. De exemplu, suma unei liste de numere se obt, ine simplu luând ca funct, ie +,
iar ca valoare init, ială 0: List.fold_left (+) 0 [1;2;3] dă rezultatul 6.

Cum tipul ’a ale rezultatului, valorilor init, iale s, i intermediare nu trebuie să fie acelas, i cu tipul ’b
al elementelor listei, cu List.fold_left se pot face prelucrări variate. De exemplu, inversarea ele-
mentelor unei liste se obt, ine cu List.fold_left (fun r e -> e :: r) [] (aplicată listei). Funct, ia
de prelucrare ia rezultatul part, ial r (partea de listă inversată) s, i adaugă elementul curent e ı̂n fat,a ei.
Astfel, dând ca valoare init, ială lista vidă, List.fold_left (fun r e -> e::r) [] [1;2;3] calcu-
lează valorile intermediare [1], [2;1], [3;2;1], returnând ultima ca rezultat. După cum se vede s, i
din tipul lui List.fold_left, funct, ia dată ca parametru ia ca prim argument rezultatul part, ial s, i ca
al doilea argument elementul listei.

List.fold_right lucrează asemănător, pornind ı̂nsă calculul de la coada listei, s, i cu o ordine
diferită pentru parametri (s, i parametrii funct, iei de prelucrare), după cum se vede din tipul funct, iei:

(’a -> ’b -> ’b) -> ’a list -> ’b -> ’b

[ e1 ; e2 ; ... en ]

r0 f r1 f r2 ... rn−1 f rn

De exemplu, List.map f lst se poate scrie List.fold_right (fun e r -> f e :: r) lst [].
List.fold_right nu e final-recursivă, deci e de preferat List.fold_left. În exemplul de mai sus,
lista ar putea fi generată s, i ı̂n ordine inversă cu List.fold_left, s, i apoi inversată.

Când compunem mai multe prelucrări de liste, e mai citet, să folosim o notat, ie alternativă pentru
aplicarea funct, iei. x |> f e echivalent cu f x, “pâlnia” sugererând că x e intrare pentru f. Astfel, o
compunere f3 (f2 (f1 x)) se poate scrie fără paranteze x |> f1 |> f2 |> f3, evident, iind ordinea
ı̂n care se aplică funct, iile: x e argument pentru f1, rezultatul fiind argumentul lui f2, etc.

De exemplu, selectăm dintr-o listă numele oras,elor cu cel put, in 200 000 de locuitori:
let orase = [("Arad",159074); ("Brasov",253200); ("Cluj",324576); ("Craiova",259506);

("Iasi",290422); ("Oradea",196367); ("Sibiu",147245); ("Timisoara",319279)]
let orase_200k = orase |> List.filter (fun (_, pop) -> pop >= 200000) |> List.map fst

Filtrarea pune o condit, ie doar pentru al doilea element al perechii, primul (_) nu contează. Selectăm
apoi primul element din pereche, cu funct, ia predefinită fst. Ea s-ar putea scrie let fst (a, _) = a .

